// File: contracts/OpenSkyPriceAggregator.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.10;

import './interfaces/IOpenSkyPriceAggregator.sol';
import '@openzeppelin/contracts/access/Ownable.sol';

interface AggregatorInterface {
    function latestAnswer() external view returns (int256);
    function latestRoundData() external view returns (
        uint80 roundId,
        int256 answer,
        uint256 startedAt,
        uint256 updatedAt,
        uint80 answeredInRound
    );
}

/**
 * @title OpenSkyPriceAggregator Contract
 * @author OpenSky Labs
 * @notice Proxy smart contract to get the price of a nft from a price aggregator, with Chainlink Aggregator
 *         smart contracts as primary option
 * - Owned by the OpenSky governance system, allowed to add aggregators for nfts, replace them
 **/
contract OpenSkyPriceAggregator is IOpenSkyPriceAggregator, Ownable {
    mapping (address => AggregatorInterface) public aggregators;

    constructor() Ownable() {}

    /**
     * @notice External function called by the OpenSky governance to set or replace price aggregators of assets
     * @param assets The addresses of the assets
     * @param _aggregators The address of the source of each asset
     **/
    function setAggregators(address[] memory assets, AggregatorInterface[] memory _aggregators) public onlyOwner {
        require(assets.length == _aggregators.length, 'INCONSISTENT_PARAMS_LENGTH');
        for (uint256 i = 0; i < assets.length; i++) {
            require(address(_aggregators[i]) != address(0), 'AGGREGATOR_CAN_NOT_BE_ZERO_ADDRESS');
            _setAggregator(assets[i], _aggregators[i]);
        }
    }

    /**
     * @notice Internal function to set the sources for each asset
     * @param asset The address of the nft collection address
     * @param aggregator The address of the aggregator
     **/
    function _setAggregator(address asset, AggregatorInterface aggregator) internal {
        aggregators[asset] = aggregator;
        emit SetAggregator(asset, address(aggregator));
    }

    /**
     * @notice Gets the nft floor price by address
     * @param asset The nft collection address
    **/
    function getAssetPrice(address asset) external view override returns (uint256) {
        if (address(aggregators[asset]) == address(0)) {
            return 0;
        }
        (, int256 answer, , , ) = aggregators[asset].latestRoundData();
        return answer > 0 ? uint256(answer) : 0;
    }
}

// File: contracts/interfaces/IOpenSkyPriceAggregator.sol



// SPDX-License-Identifier: MIT
pragma solidity 0.8.10;

interface IOpenSkyPriceAggregator {
    event SetAggregator(address indexed asset, address indexed aggregator);

    function getAssetPrice(address nftAddress) external view returns (uint256);
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}